Math 276
Calculus III

Dr. Constant J. Goutziers
Department of Mathematics, Computer Science and Statistics
goutzicj@oneonta.edu

Lesson 9

Derivatives and Integrals of Vector Functions

Initializations

```
> restart;
  with(VectorCalculus):
  BasisFormat(false):
  with(plots):
  with(oneonta):
  setoptions3d(axes=boxed):
>
```

9.1 Derivatives and Integrals of Vector Valued Functions

Examples

Example 9.1.1

Compute the unit tangent vector T to the helix defined in Example 8.1.1.

Solution

Code $\mathbf{r}(t)$ and compute

$$\mathbf{T}(t) = \frac{\mathbf{r}'(t)}{|\mathbf{r}'(t)|}$$

```
> r:=<cos(t), sin(t), t>:
  rp:=diff(r, t);
```

$$rp := \begin{bmatrix} -\sin(t) \\ \cos(t) \\ 1 \end{bmatrix} \quad \textbf{(2.1.1.1)}$$

```
> T:=rp/Norm(rp);
```

$$T := \begin{bmatrix} -\frac{1}{2}\sqrt{2}\sin(t) \\ \frac{1}{2}\sqrt{2}\cos(t) \\ \frac{1}{2}\sqrt{2} \end{bmatrix} \quad \textbf{(2.1.1.2)}$$

```
>
```

Example 9.1.2

Define r = $\langle \sin 5t, \ln 2t, t^5 \rangle$ as a Maple function and compute its derivative. Then compute the same derivative for the value $t = 4$.

Solution

```
> r:=t-><sin(5*t), ln(2*t), t^5>;
```

$$r := t \rightarrow VectorCalculus\text{:-} <,>(\sin(5\,t), \ln(2\,t), t^5) \tag{2.1.2.1}$$

Do not be confused by the unusual appearance of the output. If we ask for $\mathbf{r}(t)$, the system with give us the expected vector.

```
> r(t);
```

$$\begin{bmatrix} \sin(5\,t) \\ \ln(2\,t) \\ t^5 \end{bmatrix} \tag{2.1.2.2}$$

The derivative $\mathbf{r}'(t)$ is generated by the code.

```
> rp:=D(r)(t);
```

$$rp := \begin{bmatrix} 5\cos(5\,t) \\ \frac{1}{t} \\ 5\,t^4 \end{bmatrix} \tag{2.1.2.3}$$

The reason for coding r as a Maple function, rather than a Maple expression, is that it allows for easy evaluation of the derivative for any specific value of t. For instance $r'(4)$ is given by.

```
> D(r)(4);
```

$$\begin{bmatrix} 5\cos(20) \\ \frac{1}{4} \\ 1280 \end{bmatrix} \tag{2.1.2.4}$$

Example 9.1.3

Let $\mathbf{r}(t) = \left\langle \cos t, t^2 e^{5t}, \frac{1}{t^2+2} \right\rangle$. Compute $\int_0^{\pi} \mathbf{r}(t)\, dt$.

Solution

Code the vector $r(t)$ and evaluate the integral.

```
> r:=<cos(3*t), t^2*exp(5*t), 1/(t^2+2)>;
```

(2.1.3.1)

$$r := \begin{bmatrix} \cos(3\,t) \\ t^2\,e^{5\,t} \\ \dfrac{1}{t^2+2} \end{bmatrix} \tag{2.1.3.1}$$

```
> e1:=Int(r, t=0..Pi);
```

$$e1 := \int_0^{\pi} \begin{bmatrix} \cos(3\,t) \\ t^2\,e^{5\,t} \\ \dfrac{1}{t^2+2} \end{bmatrix} \mathrm{d}t \tag{2.1.3.2}$$

```
> e2:=value(e1);
  evalf(%);
```

$$e2 := \begin{bmatrix} 0 \\ -\dfrac{2}{125} + \dfrac{2}{125}\,e^{5\,\pi} - \dfrac{2}{25}\,e^{5\,\pi}\,\pi + \dfrac{1}{5}\,e^{5\,\pi}\,\pi^2 \\ \dfrac{1}{2}\,\arctan\left(\dfrac{1}{2}\,\pi\,\sqrt{2}\right)\sqrt{2} \end{bmatrix}$$

$$\begin{bmatrix} 0. \\ 1.153665256\;10^{7} \\ 0.8116248595 \end{bmatrix} \tag{2.1.3.3}$$

```
>
```

Example 9.1.4

Let r(t) = $\left\langle \dfrac{\sin 5\,t}{t}, t^3 \ln t, t+7 \right\rangle$. Compute $\lim_{t\to 0}$ r(t).

Solution

Code the vector and the limit.

```
> r:=<sin(5*t)/t, t^3*ln(t), t+7>:
  e1:=Limit(r, t=0);
```

$$e1 := \lim_{t\to 0} \begin{bmatrix} \dfrac{\sin(5\,t)}{t} \\ t^3\,\ln(t) \\ t+7 \end{bmatrix} \tag{2.1.4.1}$$

```
> e2:=value(e1);
```

(2.1.4.2)

$$e2 := \begin{bmatrix} 5 \\ 0 \\ 7 \end{bmatrix} \tag{2.1.4.2}$$

We could have obtained the answer immediately by using the **limit** command rather than the **Limit** command.

```
> e3:=limit(r, t=0);
```

$$e3 := \begin{bmatrix} 5 \\ 0 \\ 7 \end{bmatrix} \tag{2.1.4.3}$$

```
>
```